



# 微机原理和接口技术

## 第十一讲 定时器/计数器2

---



# 提 纲

**1. 定时器计数器概述**

**2. 定时器计数器的结构与控制**

**3. 定时器计数器的工作方式**

**4. 定时器计数器的初始化**

**5. 计数器的飞读**

**6. 定时的实现方法**

**7. 定时方式的应用**

**8. 计数方式的应用**

**9. 脉冲宽度的测量**

**10. 扩展外部中断**

**11. 实时时钟的设计**

# 提 纲

## 5. 计数器的飞读

## 定时器的飞读

要获得16位加1计数器的值，需要分别对高8位和低8位两个寄存器（THi和TLi）进行读取。若读高8位寄存器时，恰逢低8位寄存器溢出，则读取的低8位寄存器的值就会出现粗大计数误差（或称为“相位误差”）。

例：当16位计数器中的值为00FFH时

- 若先读高8位THi，得到00H，再读低8位TLi时，TLi已发生进位溢出，即此时16位计数器的值变成了0100H，所以读到TLi的值为00H，这样读取的16位计数结果变为0000H。
- 若先读低8位TLi，得到FFH，再读高8位THi时，TLi已发生进位溢出，计数器的值变成了0100H，所以读到THi的值为01H，因此得到的16位计数结果变为01FFH。

这两种结果都是不正确的



## 定时器的飞读

解决方案：采用飞读方式。

先读取THi值，再读TLi值，然后再重新读取一遍THi。若两次THi值相同，表示读取的结果正确；若不相同，则再重复上述过程。(TH0读到R0，TL0读到R1)

“飞读”子程序：

(1) 汇编程序（计数值存入R0、R1）：

```
RDT0:  MOV A,TH0           ;读入TH0到A
        MOV R1, TL0       ;读入TL0到R1
        CJNE A, TH0, RDT0  ;将第一次读取的TH0与当前的
                           ;TH0比较，不相等重读
        MOV R0, A         ;若相同，则保存TH0到R0
        RET
```

# 提 纲

## 6. 定时的实现方法



# 定时的实现方法

## (1) 软件定时

软件定时可以实现8051MCU的最短定时时间，以及长时间的定时。

- 对于只有几个机器周期的定时，通常采用若干个NOP指令来实现；
- 对于较长的定时，可以采用延时子程序实现。

## (2) 定时器定时

- 定时器方式2（8位）能够自动重装载定时初值，适合用于产生256个机器周期内的定时；
- 定时器方式1（16位）适合用于定时时间大于256小于65536个机器周期的定时。
- 更长时间的定时，将定时器的硬件定时和软件计数结合起来。

# 定时的实现方法

## 产生不同定时时间的方法

定时时间 ( $\mu\text{s}$ )	方法
几个 $\mu\text{s}$ -较长定时	软件编写
256 (max)	8位定时方式
65536 (max)	16位定时方式
较长定时	16位定时方式及软件计数

上表中假设系统晶振为12MHz，则机器周期为1 $\mu\text{s}$

- 软件延时需要CPU运行程序，因此要占用CPU的时间资源；
- 定时器不需要占用CPU时间，只有在定时器/计数器溢出时，才向CPU请求中断。定时准确、灵活性强，能有效提高微控制器的性能。



## 定时的实现方法


**例6-4：**编写一个脉冲波形产生程序，在引脚P1.0产生最高频率的周期脉冲波。能产生的最高频率是多少？该周期脉冲波形的占空比是多少？

汇编程序：

```

    ORG 0000H
LOOP: SETB P1.0    ;1
      CLR  P1.0    ;1个
      SJMP LOOP    ;2个

```



分析： END

- 在每个循环中，高电平时间为 $1\mu s$ ，低电平时间为 $3\mu s$ ，周期是 $4\mu s$ ，占空比为 $1/4=25\%$ 。因此P1.0引脚上输出的脉冲频率为 $250KHz$ 。
- 在指令SETB P1.0之后增加2条NOP指令可以使输出波形变为方波（占空比=50%），周期变为 $6\mu s$ ，频率变为 $166.7KHz$ 。



## 定时的实现方法

问题1：下段程序可在引脚P1.0产生最高频率的方波信号，请分析其最高频率和占空比是多少？（晶振12MHZ）

汇编程序：

```
ORG 0000H  
LOOP: CPL P1.0  
      SJMP LOOP  
      END
```



## 定时的实现方法

问题1：下段程序可在引脚P1.0产生最高频率的方波信号，请分析其最高频率和占空比是多少？（晶振12MHZ）

汇编程序：

```
ORG 0000H  
LOOP: CPL P1.0          ;1个 $T_M$   
      SJMP LOOP          ;2个 $T_M$   
      END
```

分析：

高低电平的时间分别为 $3\mu s$ ，周期为 $6\mu s$ ，频率 $166.67KHz$ ，占空比为50%。

# 提 纲

## 7. 定时方式的应用

## 定时方法的应用

**例6-6：** 设8051微控制器采用的晶振频率为12MHz，利用定时器T1定时，使P1.0输出周期为2ms的方波。

**分析：** 要产生周期为2ms的方波需要每1ms改变一次P1.0的电平，故定时时间为1ms，应采用工作方式1。TMOD的方式控制字应为10H；计数脉冲周期为1 $\mu$ s，所以定时初值为 $X=2^{16}-$

$1000=64536=FC18H$

**汇编程序(查询方式)：**

```
        MOV  TMOD,#10H    ;设置T1为方式1
        SETB TR1          ;启动T1定时
LOOP:   MOV  TH1,#0FCH
        MOV  TL1#18H       ;装入定时初值
        JNB  TF1,$         ;等待溢出，若TF1=0，则继续查
询等待
        CPL  P1.0          ;P1.0状态翻转，输出方波
        CLR  TF1           ;查询方式，TF1需要软件清0
        SJMP LOOP         ;重复循环
```



## 定时方法的应用

### 汇编程序(中断方式) (为什么这里没有CLR TF1? )

```
ORG    0000H
LJMP    MAIN                ;跳转到主程序
ORG    001BH                ;T1中断入口
LJMP    T1SUB               ;转T1中断服务程序
ORG    0100H
MAIN:   MOV    TMOD, #10H    ;设置T1方式1
        MOV    TH1, #0FCH
        MOV    TL1, #18H    ;设置定时初值
        SETB   EA           ;CPU中断允许
        SETB   ET1          ;T1中断允许
        SETB   TR1          ;启动T1
        SJMP   $            ;模拟主程序
```

```
ORG    0800H
T1SUB:  MOV    TH1, #0FCH    ;重载定时初值
        MOV    TL1, #18H
        CPL    P1.0         ;P1.0状态翻转, 输出方波
        RETI               ;中断返回
```



## 定时方法的应用

总结：

- 采用查询方式的程序简单，但需要CPU不断查询溢出标志，没有提高CPU的效率；（除了查询，**CPU不能干别的事**）
- 中断方式的程序编写相对复杂，但是CPU对定时器/计数器初始化后，就由硬件进行定时，CPU只在定时时间到时，响应中断执行中断服务程序即可；（**CPU可以干别的事**）
- 在实际应用程序中，通常采用中断方式设计定时程序。

# 提 纲

## 8. 计数方式的应用





## 计数方式的应用

**例6-7：**设计一个微机系统，用于记录生产流水线上每天生产的工件箱数。每箱装100个工件，因此每次计数到100个工件时，该系统要向包装机发出打包命令（输出一个高脉冲信号），使包装机执行打包动作，并推出装满工件的箱子引入空箱子。



## 计数方式的应用


**例6-7：**设计一个微机系统，用于记录生产流水线上每天生产的工件箱数。每箱装100个工件，因此每次计数到100个工件时，该系统要向包装机发出打包命令（输出一个高脉冲信号），使包装机执行打包动作，并推出装满工件的箱子引入空箱子。

**分析：**首先通过电路设计将经过流水线上的工件转换为脉冲信号，一个工件输出一个脉冲，用定时器/计数器记录工件脉冲的个数。

- 由于每100个工件为一箱，所以选择工作方式2，设置初值为156（9CH），则累计100个工件脉冲后，溢出请求中断；
- 在中断程序中发出打包信号，并完成箱数的累计等。

## 计数方式的应用

外围电路设计：选用LED光源和光敏电阻RL作为流水线上工件的检测模块。当有工件通过时，LED发出的光线受阻挡无法到达光敏电阻RL，其阻值很大而使三极管T导通输出高电平；当没有工件时，光敏电阻接收到LED光使RL变小，此时T截止而输出低电平。因此，每通过一个工件，T0端就会接收到一个正脉冲信号，由T0进行计数。





## 计数方式的应用

程序设计：计数模式，工作方式2，TMOD方式控制字为06H。用P1.0输出打包机需要的包装命令（高脉冲）；用R5和R4作为每天生产工件箱数的计数器。

```

                                ORG    0000H
汇编程序（主程序）：        SJMP    MAIN
                                ORG    000BH                ;T0中断入口
                                LJMP    COUNT                ;转向中断服务
                                ORG    0040H
MAIN:                        CLR     P1.0                ;P1.0置低电平
                                MOV     R5, #0
                                MOV     R4, #0                ;箱计数器清0
                                MOV     TMOD, #6            ;T0工作方式设置
                                MOV     TH0, #9CH            ;设置重新装载系数
                                MOV     TL0, #9CH            ;设置计数初值
                                SETB    EA
                                SETB    ET0
                                SETB    TR0
                                SJMP    $
```



# 计数方式的应用

汇编程序（中断程序）：

```
                                ORG    0800H           ;中断服务程序
COUNT:  MOV    A,R4
                                ADD    A,#1
                                MOV    R4,A           ; 记录低8位的箱数
                                MOV    A,R5
                                ADDC   A,#0           ; 记录高8位的箱数
                                MOV    R5,A           ;箱计数器加1
                                SETB   P1.0          ;输出包装机打包信号
                                MOV    R3,#10
DLY:     NOP
                                DJNZ   R3,DLY         ;使得高脉冲有一定宽度
                                CLR    P1.0          ;停止包装
                                RETI
```

# Thank you!

